Teaching Mathematics and Statistics in Sciences - IPA HU-SRB/0901/221/088
Interesting Mathematical Problems in Sciences and Everyday Life - 2011

Influenza models with Wolfram Mathematica

Diána H. Knipl & Gergely Röst
Bolyai Institute
University of Szeged, Hungary
Aradi vértanúk tere 1., Szeged, Hungary, 6720
knipl@math.u-szeged.hu, rost@math.u-szeged.hu
www.math.u-szeged.hu

1. Introduction

Finding optimal policies to minimize the mortality and morbidity of epidemic outbreaks is a top public health priority. The previous influenza pandemic and other emerging / reemerging diseases demonstrate the importance of mathematical modeling, and calls for a synergistic cooperation of epidemiologists, mathematical modelers and public health experts. The purpose of this paper is to provide an introduction to the basic principles of compartmental models and their implementation in Wolfram's Mathematica. Motivated by the 2009 A/H1N1 influenza pandemic, here we focus on how to develop advanced models for an influenza outbreak from simple building blocks using this computer algebra system. The content of this paper spans from the most basic SIR model to research level problems. Given the flexibility of the software, the models presented here are easy to modify or extend to include various intervention strategies, population structure or other features. The Manipulate tool is especially useful when someone wants to have a quick overview of the possible scenarios by changing various parameters. This way we can construct interactive and spectacular simulation tools. We hope that this work will be useful for those who are interested in infectious disease modelling, to help them to create and study their own models using Mathematica.

The basic idea of compartmental models in infectious disease modelling is that we divide our population into disjoint groups, according to a few key characteristics which are relevant to the disease under consideration. Then we model the progress of an epidemic in a large population comprising many different individuals by keeping track of the number of individuals within each subgroups, which are called compartments. For example, in many common infections, such as influenza, it makes sense to divide the population into those who are susceptible to the disease, those who are infected and those who have recovered and are immune. We can specify further compartments such as those who have been vaccinated, those who are receiving treatment, age groups, risk groups, etc. and the combinations of those to account for the heterogeneity of the population. In a dynamical model there are transition processes between the compartments that specify the rate individuals move from one compartment to the other. These are typically formulated as systems of differential equations.

A key concept in epidemic models is the basic reproduction number, denoted by R_0, defined as the average number of new infections caused by a single infected individual introduced into a wholly susceptible population over the course of the infection of this individual. In general, a disease introduced into a population will cause an epidemic if R_0 is greater than one, while the disease dies out quickly when R_0 is less than one. Thus, control measures that decrease the basic reproduction number below one may stop the epidemic even if they can not prevent all new infections. One of the most important quantity that describes the severity of an epidemic is the attack rate, which expresses the fraction of individuals who have not been avoided the infection. We can deduce a final size relation that gives a connection between the attack rate and R_0.

Influenza poses a new threat every year. Seasonal strains are related to strains that have been circulating in the

past thus a fraction of the population may have some residual immunity, while most individuals are susceptible to a novel pandemic strain. In the seasonal case the vaccination campaign typically precedes the infuenza outbreak, hence we can model this by simply assuming less susceptibles and more immunes in the initial values. However, in case of a pandemic, the vaccine may be available only in a later phase, and there is a race between the campaign and the outbreak. Modelling a delayed and continuous vaccination campaign is more challenging. Besides vaccination, antiviral treatment is an other potential mitigation strategy. A further difficulty for influenza is that not all infected individuals develop symptoms; a significant fraction of them are asymptomatic but still capable of transmitting the infection. Accordingly, we need to introduce compartments which contain the asymptomatic infected individuals. Many cases are mild enough not to be reported, hence influenza data will always be incomplete and fitting our model to real data can be problematic. It seems that model parameters for influenza are strongly age-dependent, that requires age-structured models. In particular, the contact structure between age groups has a significant effect on the outcome of the outbreak. In a real situation data are initially limited and there are uncertainties in the parameters. By careful sensitivity analyses we can examine the variation of model outputs in response to changes in input parameter values. Our models need to be constructed in a way that addresses the previous concerns.

The paper is organized as follows. Section 2 provides the analysis of the simplest SIR model and an introduction to the relevant features of Mathematica. In Section 3 we develop the SEAIR model which is the most useful for influenza, and include preseasonal vaccination and antiviral treatment as possible intervention strategies. Section 4 considers a much more complicated model with age structure and delayed vaccination campaign that is parallel to the outbreak. This reflects the real situation of the previous pandemic. Finally, we discuss some further modelling challenges in Section 5 and we explain how to use the relevant commands and tools of Mathematica that have been applied throughout the paper in an Appendix. For an introductory, but detailed text on the mathematical modelling of infectious disease, we refer the reader to O. Diekmann and J.A.P. Heesterbeek, Mathematical Epidemiology of Infectious Diseases: Model Building, Analysis and Interpretation, Wiley 2000, and Mathematical Epidemiology, Lecture Notes in Mathematics 1945, (eds. F. Brauer, P. van den Driessche and J. Wu), Springer 2008. In particular, chapters 2 and 12 of the latter by F. Brauer consists most of the contents of Section 2 and Section 3.

2. The basic SIR model

The basic compartmental models to describe the transmission of communicable diseases are originated from a sequence of papers by W.O. Kermack and A.G. McKendrick, starting from 1927. To introduce the principles of compartmental models using Mathematica and concepts such as the basic reproduction number and final size relation, we use the SIR model as a starting point. The model described in this section is a highly oversimplified special case of the general one constructed by Kermack and McKendrick that included dependence on the time elapsed since infection; however it is an important building block of more complex models. The population is divided into three classes labeled by S, I, and R. Let $S(t)$ denote the number of individuals who are susceptible to the disease at time t (measured usually in days), $I(t)$ the number of infected individuals (assuming they are able to spread the disease by contact with susceptibles), and $R(t)$ the number of individuals who have been recovered from the disease. In the case of influenza, such individuals have immunity for the same strain hence they can not be infected again during the outbreak. In a more general context, $R(t)$ may refer to the class of individuals removed from the possibility of being infected again or of spreading infection: they can either be immune, isolated or deceased. These characterizations are different from an epidemiological point of view but they result in the same model equations. To formulate our models in terms of differential equations, we assume that the number of individuals in a compartment is a differentiable function of time. As the outbreak begins, individuals are getting infected and recover, and the dynamics of this transition from one compartment to another can be described by the differential equations.

$$S'(t) = -\beta S(t) I(t), \tag{1}$$

$$I'(t) = \beta S(t) I(t) - \alpha \mathrm{I}(t), \tag{2}$$

$$R'(t) = \alpha\, I(t), \tag{3}$$

with initial conditions

$S(0) = S_0, I(0) = I_0, R(0) = R_0.$



The basic SIR model

Figure 1

This system is based on several underlying assumptions. An average individual makes βN contact per unit time which are adequate to transmit the infection to others, where $N = S(t) + I(t) + R(t)$ is the total population. Since new infection arises only when an infectious and a susceptible are in contact, the number of new infections per unit time is $\beta \mathrm{NI}(S/N) = \beta \mathrm{SI}$. Such a term called mass action incidence. We assume that infected individuals recover at rate α, thus the sojourn time in the infected compartment follows exponential distribution and the average duration of the infection is $\frac{1}{\alpha}$. Infected individuals move from the S class to I, and recovered individuals move from I to R, see Figure 1. Apart from that, there is no other entry or exit from the compartments: we assume the population is closed (no birth, natural death or migration). Furthermore, it is implicitly assumed that the population is homogenous (all individuals share the same parameters), and randomly mixing.
System (1)-(3) is non-linear, and there is no explicit analytical expression for the solutions. Nevertheless, we can give a very detailed analysis of the behaviour of the solutions. First notice that

$S'(t) + I'(t) + R'(t) = 0$

which is in accordance with our assumption

S $(t) + I(t) + R(t)$ = constant = N

for all t. Integrating (1) and (2) we can express the solutions as $S(t) = S(0)\ e^{-\beta \int_0^t I(u)\,du}$, $I(t) = I(0)\ e^{\int_0^t \beta S(u) - \alpha\,du}$, therefore if the initial values are non-negative, the solutions $S(t)$ and $I(t)$ remain non-negative for all t. The non-negativity of $R(t)$ follows form the non-negativity of $I(t)$.
From $S'(t) < 0$ we see that $S(t)$ decreases as t increases, while the number of people in class R is increasing. The following short program was written to study the properties of the epidemic curves for various values of the parameters, the population size is normalized to 1.

```
Manipulate[
 DynamicModule[{s0, r0, α, sol, s, i, r, t},
  s0 = 1 - i0; r0 = 0; α = 1 / l;
  sol = NDSolve[
     {s'[t] == -β * s[t] * i[t],
      i'[t] == β * s[t] * i[t] - α * i[t],
      r'[t] == α * i[t],
      s[0] == s0, i[0] == i0, r[0] == r0}, {s, i, r}, {t, 0, 150}];
  Plot[{s[t] /. sol, i[t] /. sol, r[t] /. sol},
   {t, 0, 30}, PlotStyle → {Darker[Green], Red, Blue},
   PlotRange → {0, 1}, PlotLegend → {"S", "I", "R"},
   LegendPosition → {0.5, -0.2}, LegendShadow → None, ShadowBorder → None,
   FrameLabel → {Style["time (in days)", Medium], Style[" ", Medium]},
   Frame → {{True, False}, {True, False}}]
 ],
 {{i0, 0.001, "I₀: infected at the beginning"},
  0, 0.1, 0.001, ImageSize → Tiny, Appearance → "Labeled"},
 {{β, 1.5, "β: transmission rate"}, 0.5, 5, 0.5,
```

```
  ImageSize → Tiny, Appearance → "Labeled"},
 {{l, 3, "1/α: infectious period (in days)"}, 1, 7,
  1, ImageSize → Tiny, Appearance → "Labeled"},
 Initialization :> (Needs["PlotLegends`"]), SaveDefinitions → True
]
```



What can be said about $I(t)$? The simulation shows that $I(t)$ is initially increasing, then after reaching a maximum it is decreasing. Is it true in general?
First let us determine the possible maximum points of $I(t)$ by examining the equation $I'(t) = 0$.

$$I'(t) = S(t)\, I(t)\, \beta - I(t)\, \alpha = 0,$$
$$S(t) = \frac{\alpha}{\beta},$$

whenever $I(t)$ is not zero. From the monotonicity of $S(t)$ we conclude that $I(t)$ has its maximum when $S(t) = \frac{\alpha}{\beta}$, and $I(t)$ can attain its maximum at most once. Also, $I(t)$ is increasing ($I'(t) > 0$), when $S(t) > \frac{\alpha}{\beta}$, and decreasing when $S(t) < \frac{\alpha}{\beta}$. The condition $S(0) > \frac{\alpha}{\beta}$ is sufficient and necessary to start an outbreak, otherwise the number of infected individuals is decreasing from the very beginning. $I(t)$ is bounded and can not oscillate, thus it must approach a limit at infinity. From (3) it follows that this limit is zero.

2.1. Basic reproduction number

The basic reproduction number, denoted by R_0, is one of the most important parameters of an epidemic. R_0 expresses the expected number of secondary infections generated by a single infectious individual introduced into a fully susceptible population. This quantity determines weather a disease can invade a population. For the SIR model, initially an infected individual generates $\beta S(0)$ infections per unit time, and given that the duration of the infectious period is $\frac{1}{\alpha}$, we obtain $R_0 = S(0)\frac{\beta}{\alpha}$. Recall that for an outbreak to start we had the condition $S(0) > \frac{\alpha}{\beta}$, which is equivalent with $R_0 > 1$. Thus, the reproduction number is a threshold quantity. To control the

disease, the reproduction number should be decreased below one.

2.2. Final size relation

Since $S(t)$ is decreasing but remains non-negative , the limit $\lim_{t\to\infty} S(t) = : S_\infty$ exists. Finding this limit provides important information, because this quantity expresses how many susceptibles avoided the infection during the course of the outbreak, or equivalently what was the total number of infections. From $(I(t) + S(t))' = -\alpha I(t)$ it follows $I_\infty = 0$. Integrating this equation we have $S_\infty - S(0) - I(0) = -\alpha \int_0^\infty I(u)\,du$. Taking the limit in $S(t) = S(0)e^{-\beta \int_0^t I(u)\,du}$, we obtain $S(\infty) = S(0)e^{-\beta \int_0^\infty I(u)\,du}$, or $\log S_\infty = \log S(0) - \beta \int_0^\infty I(u)\,du$, and finally $\log S_\infty = \log S(0) - \beta \frac{I(0)+S(0)-S_\infty}{\alpha}$.

Assuming that $I(0)$ is small and neglecting it, we have the final size relation

$$\log\left(\frac{S_\infty}{S(0)}\right) \approx -\frac{\beta}{\alpha}\,(S(0) - S_\infty),$$
$$\log\left(\frac{S(0)}{S_\infty}\right) \approx R_0\left(1 - \frac{S_\infty}{S(0)}\right).$$

The relation between the final size of the epidemic and the basic reproduction number is plotted next.



2.3. First integral

First integrals (invariants) carry important information about the behaviour of nonlinear systems. In our case, we are looking for a first integral $V(S, I) : \mathbb{R}^2 \to \mathbb{R}$, such that V is constant along solutions (i.e. solutions live on the level sets of V). Let us look for the first integral in the following form: $V(S, I) = S + I - c \log S$. After differentiating with respect to t and using the differential equations, one easily gets that along solutions $\frac{dV}{dt}$ equals to 0 if and only if $c = \frac{\alpha}{\beta}$. So $V(S(t), I(t)) = S(t) + I(t) - \frac{\alpha}{\beta} \log S(t)$ is a first integral. Since $V(S(t), I(t)) = \text{constant} = C$, we can deduce the final size relation from $V(S(0), I(0)) = V(S_\infty, I_\infty)$ which gives $S(0) + I(0) - \frac{\alpha}{\beta} \log S(0) = S_\infty + I_\infty - \frac{\alpha}{\beta} \log S_\infty$, the equivalent of $R_0\left(1 - \frac{S_\infty}{S(0)}\right) \approx \log \frac{S(0)}{S_\infty}$ after ignoring $I(0)$. By the first integral, we can determine the peak size of the epidemic. Since $I(t)$ attains its maximum when $S(t) = \frac{\alpha}{\beta}$, we obtain $I_{\max} = S(0) + I(0) - \frac{\alpha}{\beta} \log S(0) - \frac{\alpha}{\beta} + \frac{\alpha}{\beta} \log \frac{\alpha}{\beta}$. Mathematica can plot the level curves of V on the S-I plane, thus we can have a clear picture of the phase curves of the system, as can be seen below. Since S is decreasing, as time elapses solutions move to the left on the S-I phase plane along the level curves.



3. Influenza models - asymptomatic infection, vaccination and antiviral treatment

3.1. SEAIR model

To include two important aspects of influenza, we extend the basic SIR model. There is an incubation period between infection and the development of the disease so that an infected person becomes infectious. Thus, we introduce an intermediate compartment E. A significant fraction of people who have been infected never develop symptoms, so they will never be detected. However, going through an asymptomatic infection, they are capable of transmitting the infection. Thus the new model contains the compartments S, E, I, A and R. Upon adequate contact with an infective, susceptibles move into the compartment E. After the incubation period (which has length $\frac{1}{\mu_E}$), they develop symptoms with probability p, or become asymptomatic infected with probability $1 - p$. Asymptomatic infected individuals are less infectious by a factor δ. The recovery rates are μ_I and μ_A, respectively. See Figure 2 for the flow chart of the SEAIR model, the arrows indicate the movement of individuals between compartments.



The chart flow on the SEAIR − model

The model equations take the form

$$S'(t) = -\beta\, S(t)\,(\delta\, A(t) + I(t)), \tag{4}$$

$$E'(t) = \beta\, S(t)\,(\delta\, A(t) + I(t)) - \mu_E\, E(t), \tag{5}$$

$$I'(t) = p\,\mu_E\, E(t) - \mu_I\, I(t), \tag{6}$$

$$A'(t) = (1-p)\,\mu_E\, E(t) - \mu_A\, A(t), \tag{7}$$

$$R'(t) = \mu_A\, A(t) + \mu_I\, I(t), \tag{8}$$

with initial conditions

$S(0) = S_0, E(0) = E_0, I(0) = I_0,\ A(0) = A_0, R(0) = R_0.$

For a single influenza outbreak, we can neglect natural death, birth and migration, which takes place on a much longer time scale. The equations do not account for the disease induced deaths, but once the mortality rate is known, the number of the fatal cases can be easily computed from the total number of infections.

Taking into account the average times spent is compartments, we can express the expected number of secondary infections generated by a single infective in a susceptible population as

$R_0 = \beta\, S(0)\left(\frac{p}{\mu_I} + \delta\,\frac{1-p}{\mu_A}\right).$

Parameter	Description	Value
β	transmission rate	0 – 0.00003
$\frac{1}{\mu_I}$	duration of infectious period (symptomatic)	2.85 days
$\frac{1}{\mu_A}$	duration of infectious period (asmyptomatic)	4.1 days
$\frac{1}{\mu_E}$	latency period	1.25 days
δ	reduction of infectiveness for asymptomatic infections	0.071
$S(0)$	susceptibles initially	60 000 – 100 000 / 100 000
$E(0)$	latent initially	0 – 150 / 100 000
p	probability of developing symptoms	0.5 – 0.7
N	population size	100 000
R_0	basic reproduction number	$S(0)\ \beta\left(\frac{(1-p)\,\delta}{\mu_A} + \frac{p}{\mu_I}\right)$

Typical model parameters

Table 1 summarizes the parameter ranges used in the following simulation. Since a fraction of population is expected to have some residual immunity against circulating seasonal strains, we let $S(0)$ vary on a wide range.



3.2. Preseasonal vaccination

The most effective control strategy against influenza is vaccination. For seasonal influenza, vaccination campaigns precede the outbreak, so from a modeling point of view to incorporate vaccination we simply remove the immunized population from the susceptible compartment and start the model with a lower value of $S(0)$. In the presence of an intervention strategy, the reproduction number is modified, and called control reproduction number, denoted by R_c. R_c can be calculated analogously as R_0 in the absence of vaccination. To control the outbreak, R_c should be less than one. However, any reduction in the reproduction number mitigates the severity of the epidemic. Since vaccination is not 100 % effective, in the interactive simulation next we introduce an additional parameter q which expresses the chance that vaccination is successful. The initial values are given so that $R_c \approx 1.4$.



3.3. Antiviral treatment

Another possible intervention strategy against influenza is antiviral treatment, specially when for some reason there is no vaccine available. Antiviral treatment reduces the duration of the infection to $\frac{1}{\mu_T}$, and also the infectivity of a treated person by a factor σ. Let τ be the rate symptomatic infected individuals receive treatment and θ is the rate of relapse. An unsuccessfully treated person can not be treated again, which menans this person moves to a new class denoted by I_U. The compartment of treated individuals is denoted by I_T. Then we have the following model:

$$S'(t) = -\beta S(t)\,(\delta A(t) + I(t) + \sigma I_T(t) + I_U(t)), \tag{9}$$

$$E'(t) = \beta S(t)\,(\delta A(t) + I(t) + \sigma I_T(t) + I_U(t)) - \mu_E E(t), \tag{10}$$

$$I'(t) = p\,\mu_E E(t) - \mu_I I(t) - \tau\,\mathrm{I}(t), \tag{11}$$

$$I_T'(t) = \tau I(t) - \theta I_T(t) - \mu_\mathrm{T}\,\mathrm{I_T}(t), \tag{12}$$

$$I_U'(t) = \theta I_T(t) - \mu_\mathrm{I}\,\mathrm{I_U}(t), \tag{13}$$

$$A'(t) = (1-p)\,\mu_E E(t) - \mu_A A(t), \tag{14}$$

$$R'(t) = \mu_A A(t) + \mu_I I(t) + \mu_\mathrm{T}\,\mathrm{I_T}(t) + \mu_\mathrm{I}\,\mathrm{I_U}(t), \tag{15}$$

whit initial conditions

$S(0) = S_0$, $E(0) = E_0$, $I(0) = I_0$, $I_T(0) = I_{T0}$, $I_U(0) = I_{U0}$, $A(0) = A_0$, $R(0) = R_0$.

The control reproduction number for this model can be computed as :

$$R_c = \beta S(0)\left(p\left(\frac{1}{\mu_I+\tau}+\frac{\tau}{\mu_I+\tau}\left(\frac{\sigma}{\theta+\mu_T}+\frac{\theta}{\theta+\mu_T}\frac{1}{\mu_I}\right)\right)+\frac{(1-p)\,\delta}{\mu_A}\right)$$



3.4. Other interventions

There are other possible intervention strategies, such as lowering the contact number (and thus β) by school closures or campaigning to avoid crowded places. Prophylaxis can be given to strategic personnel or close contacts of symptomatic infectives. Such measures or the combinations of them can be incorporated into the SEAIR model. For various simple models with vaccination and antiviral treatment, we refer to Arino et al. 2006, 2008, Brauer 2008.

4. A pandemic model with age structure and delayed vaccination campaign

In the spring of 2009 in Mexico, a new influenza strain appeared and spread quickly all over the world. Vaccination campaigns started all around the world as a primary mitigation strategy against the first wave of the 2009 A(H1N1)v pandemic, though in several countries the vaccine became available only in a later phase of the pandemics or with limited supplies. In a pandemic situation it is typical that there is an ongoing race between the vaccination campaign and the dynamics of the outbreak, which is a more challenging modeling problem than preseasonal vaccination.

We developed a compartmental model based on the SEIR model (S: susceptible, E: exposed, I: infected, R: recovered) incorporating three important aspects of pandemic influenza to make the model more realistic.

i) Age structured models are necessary for multiple reasons: various age groups have different contact profiles thus playing different roles in transmitting the disease, and several important parameters are age dependent. We introduced age structure with five age groups (0-9, 10-19, 20-39,
40-64 and 65+ years old), where the contacts between age groups are derived from the European survey Mossong et al. 2008. The importance of age specificity has been addressed in several studies (Longini & Halloran 2005, Medlock & Galvani 2009).
ii) According to serological studies, it takes about 14 days for the human body to develop antibodies after vaccination to acquire immunity. During this intermediate period an individual might contract the disease. This time delay can be significant when vaccination is given during the outbreak.
ii) Optimal distribution of vaccines among different groups has been studied. Here we focus on the effect of the scheduling of immunization of age groups.

We compare five vaccination strategies which differ in prioritizing the age groups in the timing of their vaccination. We targeted a 60 % vacination coverage by the end of a three months vaccination campaign. The model has been discussed in detail in Knipl & Röst 2011, in the sequel we outline the main results and provide a Mathematica code that can be used to simulate various scenarios.

4.1. Model description



Flow chart without age structure

Figure 3

Since we model a single pandemic wave, natural death, birth, migration are ignored. We assume no pre-existing immunity in the first four age groups, and 20 % reduction in susceptibility in the elder age group. Since the latent period is relativily short (1.25 days), we neglect the small probability of someone receiving the vaccine while bleing in class E. Vaccination is only administered for individulas in the class S until we reach the targeted 60 % population levele coverage. Vaccinees move into the class W for an itermediate period during that infection is still spossible. After 14 days they become either immune with probability q and move into class R_W or if the vaccine was ineffective, they move into S_V meaning that they are still susceptible to the disease despite having been vaccinated. Such individulas will not receive the vaccine again, but still can contract the disease. It is assumed in the baseline scenario that the same epidemiological parameters apply to these individulas as to the non-vaccinated ($\mu_{E_V} = \mu_E$ and $\mu_{I_V} = \mu_I$, the durations of the incubation and the infective period), and we assume that vaccinated infected individuals (I_V) are less intefctious by the reduction parameter δ). For values of the parameters of the baseline scenario, see Table 2.

Parameter	Description	Value
$\frac{1}{\mu_E}$	latent period	1.25 days
$\frac{1}{\mu_I}$	infectious period	3 days
q_i, $i = 1, \ldots 4$	vaccine efficiency for 0 – 65 years old	0.8
q_5	vaccine efficiency for 65 + years old	0.6
$\beta_{i,j}$	transmission rate	see 4.3.
$\frac{1}{\mu_W}$	time to develop antibodies	14 days
δ	reduction in infectiousness	0.75

The key Model parameters. Sources : Balcan et al. 2009, Basta et al. 2008, Nichol 1998

We have 10 different classes for each age groups, and 5 age groups, so overall there are 50 compartments. The corresponding system of differential equations follow, where the upper index i, $i = 1,\ldots 5$ denotes the age groups, $\lambda^i(t) = \sum_{j=1}^{5} \beta_{i,j} \left(I^j(t) + \delta I_V^j(t) \right)$ denotes the force of infection, and $V^i = V^i(t)$ is the prescribed vaccination rate function determined by the specific strategy.

$$\mathbf{S^{i\,\prime}(t) = -S^i(t)\,\lambda^i(t) - V^i(t),} \tag{16}$$

$$\mathbf{E^{i\,\prime}(t) = \left(S^i(t) + W^i(t)\right)\lambda^i(t) - \mu_{E^i}\,E^i(t),} \tag{17}$$

$$\mathbf{I^{i\,\prime}(t) = \mu_{E^i}\,E^i(t) - \mu_{I^i}\,I^i(t),} \tag{18}$$

$$\mathbf{R^{i\,\prime}(t) = \mu_{I^i}\,I^i(t),} \tag{19}$$

$$\mathbf{W^{i\,\prime}(t) = V^i(t) - W^i(t)\,\lambda^i(t) - \mu_W\,W^i(t),} \tag{20}$$

$$\mathbf{S_V^{i\,\prime}(t) = (1 - q_i)\,\mu_W\,W^i(t) - S_V^i(t)\,\lambda^i(t),} \tag{21}$$

$$\mathbf{E_V^{i\,\prime}(t) = S_V^i(t)\,\lambda^i(t) - \mu_{E_V^i}\,E^i(t),} \tag{22}$$

$$\mathbf{I_V^{i\,\prime}(t) = \mu_{E_V^i}\,E^i(t) - \mu_{I_V^i}\,I^i(t),} \tag{23}$$

$$\mathbf{R_V^{i\,\prime}(t) = \mu_{I_V^i}\,I^i(t).} \tag{24}$$

Our model starts at $t = 0$, time is measured in days. We assume that the initial number of infected individuals is low. The time T refers to the delay in start of the campaign, meaning that the vaccination starts on day T . In the baseline scenario it takes 90 days to reach the targeted 60% vaccination coverage. Vaccination strategies are compared to each other by two outcome measures: the overall attack rates and mortality. Attack rate is the cumulative incidence of the infection during the whole time period of the pandemic wave.

4.2. Age structure

The age distribution of the population is based on Eurostat 2006 (see Table 3). We can observe the contact structure of the population in the contact matrix C (Table 4), where the elements $c_{i,j}$ represent the number of contacts of an individual in age group i has with individuals in age group j, is derived from Mossong et al. 2008 by applying an averaging and symmetrization method.

N^1	N^2	N^3	N^4	N^5
10 500	12 000	28 500	32 500	16 500

Age distribution of the population per 100 000 (source: Eurostat 2006)

$$C=\begin{pmatrix} 5,3580 & 1,0865 & 3,0404 & 2,4847 & 0,8150 \\ 0,9507 & 10,2827 & 2,8148 & 3,6215 & 0,7752 \\ 1,1201 & 1,1852 & 6,5220 & 4,1938 & 0,9016 \\ 0,8027 & 1,3372 & 3,6776 & 5,2632 & 1,3977 \\ 0,5187 & 0,5638 & 1,5573 & 2,7531 & 2,0742 \end{pmatrix}$$

Contact matrix for the five age groups, constructed by the data of Mossong et al. 2008 and Eurostat 2006

Table 3

Age specific contact rates can be converted to age specific transmission rates $\beta_{i,j}$ as follows. The average number of contacts made by a member of the age group i with a member of age group j is $c_{i,j}$. At time t, $\frac{S^j(t)}{N^j}$ is the proportion of susceptibles, $\frac{S_V^j(t)}{N^j}$ is the proportion of vaccinated susceptibles, $\frac{W^j(t)}{N^j}$ is the proportion of those who have already been vaccinated but not yet protected in age group j. From this we obtain that the rates of infections in age group j by individuals in age group i, and letting β be the transmission parameter which involves the normalization of the contacts to unit time and the infectiousness of the virus, $\beta_{i,j} = \beta \frac{c_{i,j}}{N^j}$ gives the correct parameters used in the differential equations.

4.3. Reproduction number

The elements of the next generation matrix N are given by the formula $(N)_{i,j} = n_{i,j} = \left[\frac{\beta_{i,j} S_0^j}{\mu_I^i}\right]_{i,j}$, expressing the number of infections in age group j generated by an infected individual of age group i during the course of its infection in the early phase of the pandemic. The reproduction number is the largest eigenvalue of the next generation matrix (see Diekmann 2010) . We can scale the matrix by β to achieve any value of the reproduction number. In the baseline scenraio, we consider R_0=1.4, which corresponds to β = 0.0334.

4.4. Vaccination strategies

We evaluated five different vaccination scheduling, which are described below.

'A' - Conventional strategy
It describes a common vaccination campaign which are applied in many countries in epidemic situations. It consists of three phases:

Phase 1: 42 days, vaccination of high risk groups, elder people, emergency and health care personel, workers of critical infrastructure facilities.
Phase 2: 18 days long, vaccination of children under 19.
Phase 3: In the last 30 days, vaccination is given to the general population such that we achieve the 60 % coverage by the end of this phase in each age groups.

'B' - Uniform strategy
This is a universal vaccination strategy, when there are no prioritized age groups, so we assume that vaccination is completely random and 0.667% of each age group is vaccinated daily, throughout 90 days.

'C' - Elderly first
Phased vaccination of elder people (older than 65) first up to 60 % coverage (15 days) before vaccine is delivered to the other part of the population (75 days).

'D' - Children first

Phased vaccination of children (younger than 19) first up to 60 % coverage (20 days) before vaccine is delivered to the other part of the population (70 days).

'E' - By contacts
Here we take adventage of the full contact structure of the five age groups, and vaccinate them in five phases according to the decreasing order of their total contact numbers (according to the contact matrix).
Phase 1: 10-19 years old, 11 days
Phase 2: 20-39 years old, 26 days
Phase 3: 0-9 years old, 10 days
Phase 4: 40-64 years old, 29 days
Phase 5: 65 and elder, 15 days

4.5. Main results

We have evaluated and compared the above described five strategies for various delays in start of the vaccination. Our main outcome measures are the (age specific) attack rates.



Total attack rates for the 5 strategies with various delays in start of the vaccination campaign

Figure 4

Figure 4 shows the attack rates for the five strategies. We consistently obtained the lowest attack rates by strategy *E*, followed by *D*, *B*, *A* and *C*, for all values of *T*. However, an unbalanced age specific mortality pattern may cause that not necessarily the lowest attack rate corresponds to the lowest fatality rate.



Age specific attack rates for the 5 strategies (A-E) and in the absence of vaccination (0). The colors indicate the increases inthe attack rates for longer delays in start of the campaign (T = 0, T = 25 and T = 50).

Figure 5

Younger age groups tend to have higher contact numbers than others, which leads to the interesting phenomenon that sometimes an age group benefits from having vaccinated later, since the early vaccinated key groups can

provide indirect protection. Such results are detailed in Knipl and Röst 2011, together with a comprehensive sensitivity analysis. Next we present a Mathematica code that simulates this age structured SEIR-model model with five age groups. It includes the full structure of vaccination (as in Figure 3), where the functions that describe vaccination are temporarily set according to a uniform strategy (we do not give priority to any age group). One can easily define any other vaccination strategy, calculate the corresponding vaccination functions and incorporate them into the code, and then run simulations to explore the possible outcomes.

```
RunModel[q_, T0_, brn_, e30_] :=
 Module[
  {sol, NGMu, β, AR, ARALL, e0, s0, l, h, T, beta, δ, vf, s, s1, s2, s3, s4, s5, w, w0,
   w1, w2, w3, w4, w5, e, e1, e2, e3, e4, e5, ii, i0, ii1, ii2, ii3, ii4, ii5, r,
   r0, r1, r2, r3, r4, r5, rw, rw0, rw1, rw2, rw3, rw4, rw5, sv, sv0, sv1, sv2, sv3,
   sv4, sv5, ev, ev0, ev1, ev2, ev3, ev4, ev5, iv, iv0, iv1, iv2, iv3, iv4, iv5, rv,
   rv0, rv1, rv2, rv3, rv4, rv5, model, t, c, ps, ag = 5, μw = 1 / 14, μe = 1 / 1.25,
   μi = 1 / 3, μev = 1 / 1.25, μiv = 1 / 3, n = {10 500, 12 000, 28 500, 32 500, 16 500}},
  c =
   {{5.358, 1.0865, 3.0404, 2.4847, 0.815},
    {0.9507, 10.2827, 2.8184, 3.6215, 0.7752},
    {1.1201, 1.1852, 6.522, 4.1938, 0.9016},
    {0.8027, 1.3372, 3.6776, 5.2632, 1.3977},
    {0.5187, 0.5638, 1.5573, 2.7531, 2.0742}};
  ps = Total[n];
  w0 = ConstantArray[0, ag];
  i0 = ConstantArray[0, ag];
  r0 = ConstantArray[0, ag];
  rw0 = ConstantArray[0, ag];
  sv0 = ConstantArray[0, ag];
  ev0 = ConstantArray[0, ag];
  iv0 = ConstantArray[0, ag];
  rv0 = ConstantArray[0, ag];
  s = {s1[t], s2[t], s3[t], s4[t], s5[t]};
  w = {w1[t], w2[t], w3[t], w4[t], w5[t]};
  e = {e1[t], e2[t], e3[t], e4[t], e5[t]};
  ii = {ii1[t], ii2[t], ii3[t], ii4[t], ii5[t]};
  r = {r1[t], r2[t], r3[t], r4[t], r5[t]};
  rw = {rw1[t], rw2[t], rw3[t], rw4[t], rw5[t]};
  sv = {sv1[t], sv2[t], sv3[t], sv4[t], sv5[t]};
  ev = {ev1[t], ev2[t], ev3[t], ev4[t], ev5[t]};
  iv = {iv1[t], iv2[t], iv3[t], iv4[t], iv5[t]};
  rv = {rv1[t], rv2[t], rv3[t], rv4[t], rv5[t]};
  model = {s, w, e, ii, r, rw, sv, ev, iv, rv};
  e0 = {0, 0, e30, e30, 0};
  s0 = n - e0;
  (*Here you can set all the four
   parameters necessary to describe a vaccination strategy.*)
  δ = 0.75; (*Previously unsuccessfully vaccinated,
  infected individuals are considered to be less infective.*)
  l = {90, 90, 90, 90, 90}; (*Determine the length of the
   duration of vaccination in each age group. Note that
   vaccination in various age groups can run parallel, as well.*)
  h = Table[60 000 / l[[i]], {i, 1, ag}];
  (*Set the amount of vaccine provided for the age groups per day.*)
  T = T0 + {0, 0, 0, 0, 0};  (*List 'T' is for
   scheduling: how many days after the outbreak should vaccination
     begin in the age groups. T0 is the number of days elapsed
```

```
    after the outbreak until the global start of the campaign.*)
NGMu = Table[(c[[i, j]] / n[[j]]) * s0[[j]] / μi, {i, 1, ag}, {j, 1, ag}];
β = brn / Max[Map[Re, Eigenvalues[NGMu]]];
beta = Table[β * c[[i, j]] / n[[j]], {i, 1, ag}, {j, 1, ag}];
vf =
 Table[h[[i]] * UnitStep[t - T[[i]]] * UnitStep[T[[i]] - t + l[[i]]], {i, 1, ag}];
sol =
 NDSolve[
  Union[
   Table[
    D[s, t][[i]] ==
     -s[[i]] * Sum[beta[[j, i]] * (ii[[j]] + δ * iv[[j]]), {j, 1, ag}] - vf[[i]]
    , {i, 1, ag}],
   Table[
    D[w, t][[i]] == vf[[i]] -
      w[[i]] * Sum[beta[[j, i]] * (ii[[j]] + δ * iv[[j]]), {j, 1, ag}] - μw * w[[i]]
    , {i, 1, ag}],
   Table[
    D[e, t][[i]] == (s[[i]] + w[[i]]) *
       Sum[beta[[j, i]] * (ii[[j]] + δ * iv[[j]]), {j, 1, ag}] - e[[i]] * μe
    , {i, 1, ag}],
   Table[
    D[ii, t][[i]] == e[[i]] * μe - ii[[i]] * μi
    , {i, 1, ag}],
   Table[
    D[r, t][[i]] == ii[[i]] * μi
    , {i, 1, ag}],
   Table[
    D[rw, t][[i]] == q * μw * w[[i]]
    , {i, 1, ag}],
   Table[
    D[sv, t][[i]] == (1 - q) * μw * w[[i]] -
      sv[[i]] * Sum[beta[[j, i]] * (ii[[j]] + δ * iv[[j]]), {j, 1, ag}]
    , {i, 1, ag}],
   Table[
    D[ev, t][[i]] == sv[[i]] *
       Sum[beta[[j, i]] * (ii[[j]] + δ * iv[[j]]), {j, 1, ag}] - μev * ev[[i]]
    , {i, 1, ag}],
   Table[
    D[iv, t][[i]] == μev * ev[[i]] - μiv * iv[[i]]
    , {i, 1, ag}],
   Table[
    D[rv, t][[i]] == μiv * iv[[i]]
    , {i, 1, ag}],
   Table[(s[[i]] /. t → 0) == s0[[i]], {i, 1, ag}],
   Table[(ii[[i]] /. t → 0) == i0[[i]], {i, 1, ag}],
   Table[(e[[i]] /. t → 0) == e0[[i]], {i, 1, ag}],
   Table[(r[[i]] /. t → 0) == r0[[i]], {i, 1, ag}],
   Table[(w[[i]] /. t → 0) == w0[[i]], {i, 1, ag}],
   Table[(rw[[i]] /. t → 0) == rw0[[i]], {i, 1, ag}],
   Table[(sv[[i]] /. t → 0) == sv0[[i]], {i, 1, ag}],
   Table[(iv[[i]] /. t → 0) == iv0[[i]], {i, 1, ag}],
   Table[(ev[[i]] /. t → 0) == ev0[[i]], {i, 1, ag}],
   Table[(rv[[i]] /. t → 0) == rv0[[i]], {i, 1, ag}]
  ]
  , Flatten[Table[#[[0]] & /@ variable,
```

```
      {variable, {s, e, ii, r, w, rw, sv, iv, ev, rv}}]], {t, 0, 250}];
  AR = Flatten@Table[1 - (((s[[i]] /. sol) /. t → 200) +
        ((sv[[i]] /. sol) /. t → 200) + ((rw[[i]] /. sol) /. t → 200) +
        ((w[[i]] /. sol) /. t → 200)) / n[[i]], {i, 1, ag}];
  ARALL = Total[AR * n] / Total[n];
  {(Table[#[[0]] & /@ variable, {variable, {s, e, ii, r, w, rw, sv, iv, ev, rv}}] /.
      sol)[[1, {3, 8}]], AR, ARALL, h, l, T}
 ]
vaccinatedPopulation[h_, l_, T_, t_, ag_] := Table[
  Piecewise[{{0, t <= T[[i]]}, {h[[i]] (t - T[[i]]), T[[i]] + l[[i]] >= t > T[[i]]},
    {h[[i]] l[[i]], t > T[[i]] + l[[i]]}}], {i, 1, ag}]
plotColors = {{Thick, RGBColor[0., 0.66, 0.05]},
   {Thick, RGBColor[1, 0.58, 0.066]}, {Thick, RGBColor[0.25, 0., 0.8]},
   {Thick, RGBColor[1., 0., 0.]}, {Thick, RGBColor[0., 0., 0.]}};
tableColors = {RGBColor[0.4, 1, 0.66], RGBColor[1, 0.81, 0.53],
   RGBColor[0.6, 0.7, 1.], RGBColor[1., 0.6, 0.6], RGBColor[0.75, 0.75, 0.75]};
ageGroups = {"0-9", "10-19", "20-39", "40-64", "65-"};
```

```
Manipulate[
 DynamicModule[
  {solution, ar, overallAR, h, l, T, attackRates,
   t, vacc = 0.6, ag = 5, n = {10 500, 12 000, 28 500, 32 500, 16 500},
   ii, ii1, ii2, ii3, ii4, ii5, iv, iv1, iv2, iv3, iv4, iv5},
  {solution, ar, overallAR, h, l, T} = RunModel[q, T0, brn, e30];
  attackRates = Round[#, 0.01] & /@ (100. Flatten[{ar, overallAR}]);
  ii = {ii1, ii2, ii3, ii4, ii5};
  iv = {iv1, iv2, iv3, iv4, iv5};

  Plot[
   Evaluate[
    Table[(solution[[1, i]][t] + solution[[2, i]][t]), {i, 1, ag}]],
   {t, 0, 150}, PlotRange → {0, 1200}, PlotStyle → plotColors,
   ImageSize → {300, 250}, AspectRatio → 0.9, AxesLabel -> {"t (days)", "cases"},
   PlotLabel → Grid[{{Style["AR=", Large, 16], Style[overallAR , Large, 16]} }],
   PlotLegend → {"0-9", "10-19", "20-39", "40-64", "65+"},
   LegendBorder → White, LegendShadow → None, LegendPosition -> {0.6, -0.4}]

 ],
 Grid[
  {
   {"vaccine efficacy", Control[
     {{q, 0.6, ""}, 0.5, 0.9, 0.01, ImageSize → Small, Appearance → "Labeled"}]},
   {"starting time of vaccination", Control[
     {{T0, 40, ""}, 0, 50, 5, ImageSize → Small, Appearance → "Labeled"}]},
   {"basic reproduction number", Control[{{brn, 1.55, ""}, 1.2,
      1.7, 0.01, ImageSize → Small, Appearance → "Labeled"}]},
   {"infected initially", Control[{{e30, 7, ""}, 0, 20, 1,
      ImageSize → Small, Appearance → "Labeled"}]}
  }, Alignment → Left
 ], Initialization :> (Needs["PlotLegends`"]), SaveDefinitions → True
]
```



4.6. Application to the first wave of A/H1N1 in Hungary

Here we briefly illustrate how this 50-compartment model can be applied in a real life situation. Our example is the first wave of A/H1N1 in Hungary. Epidemic curves were reconstructed using the public reports of the National Center of Epidemiology (www.oek.hu). For the simulations, we fixed the epidemiological parameters as in Table 2, employed publicly available vaccination data (www.jarvany.hu) and performed a grid search with respect to the basic reproduction number and the reduction of contacts during holidays to find the best fit by means of ordinary least square method. The result can be seen in Figure 6, where the first day corresponds to August 24 and R_0~1.3. The vaccination started on day 36, the red curve was made about day 80, so that was a prediction for the part of the epidemic curve after that day. The model accurately predicted the peak size of the outbreak and also that the number of infections would not increase after Christmas break.



Hungarian epidemic curve and model prediction

5. Further advanced models

There are many more features that one can easily incorporate into the models and the codes presented here. For example, we assumed in all the models that the duration of latent and infectious periods follow exponential distributions. By the so-called linear chain trick, by artificially dividing a period into smaller subperiods, having exponential distribution in each, we can model gamma-distributed infectious or latency periods, and still having ordinary differential equations. The 14 days period that required to develop immunity after vaccination was also assumed to follow exponential distribution, however the fixed 14 days assumption seems more realistic. By assuming that for every individual it takes exactly 14 days to acquire immunity, we obtain a delay differential equation, which is mathematically rather complicated. In particular, for class W we have the equation

$$W'(t) = V(t) - \lambda(t) - V(t-14)\ e^{-\int_{t-14}^{t}\lambda(s)\,ds},$$

where we omitted the indices and λ(t) is the infection term that expresses the rate recent vaccinees contract the infection, and may depend on the state of many other compartments. This is a differential equation with discrete and distributed delays. Another modeling approach also leads to differential equations with distributed and discrete delays, namely the age since infection model of antiviral treatment, which takes into account the window of opportunity for initiating treatment and the dependence of the treatment rate on the time elapsed since infection. This model family have been studied in detail in Alexander et al. 2007, 2008 Moghadas et al. 2008, 2009, where the possibility of emergence of a resistant strain and antiviral prophylaxis have also been considered. The command 'NDSolve' we used in this paper can handel differential equations with discrete delays, i.e. equations of the form

$$x'(t) = f(x(t-T_1),\ x(t-T_2),\ \ldots,\ x(t-T_k)).$$

Then, instead of the initial value x[0] == c , the initial history function x[t /; t ≤ 0] == h[t] should be specified in the 'NDSolve' command. At the moment, 'NDSolve' does not seem to be able to handle disributed delays. However, the integral terms that represent the distributed delay can be approximated by their Riemann sums, for example $\int_{t-14}^{t}\lambda(s)\,ds \approx \sum_{j=0}^{13}\lambda(t-j)$. This way we can construct an approximative equation

$$W'(t) = V(t) - \lambda(t) - V(t-14)\,e^{-\sum_{j=0}^{13}\lambda(t-j)}$$

with several discrete delays that can be treated by 'NDSolve'.

Acknowledgement

GR was partially supported by the Hungarian Research Fund grant OTKA K75517, the Bolyai Research Scholarship of the Hungarian Academy of Sciences, and the TÁMOP - 4.2 .2/08/1/2008 - 0008 program of the Hungarian National Development Agency.

References

Alexander ME, Bowman CS, Feng Zh, Gardam M, Moghadas SM, Röst G, Wu J, Yan P 2007 Emergence of drug-resistance: implications for antiviral control of influenza pandemic P Roy Soc B - Biol Sci 274:(1619) 1675-1684

Alexander ME, Moghadas SM, Röst G, Wu J 2008 A delay differential model for pandemic influenza with antiviral treatment, Bulletin of Mathematical Biology 70(2), 382-397

Arino J, Brauer F, van den Driessche P,Watmough J, Wu, J 2006 Simple models for containment of a pandemic J R Soc Interface 3(8) 453-457

Arino J, Brauer F, van den Driessche P,Watmough J, Wu, J 2008 A model for influenza with vaccination and antiviral treatment J Theor Biol 253, 118-130

Balcan D et al. 2009 Seasonal transmission potential and activity peaks of the new influenza A(H1N1): a Monte Carlo likelihood analysis based on human mobility BMC Medicine 7(45) doi:10.1186/1741-7015-7-45

Basta NE, Halloran EM, Matrajt L, Longini IM Jr. 2008 Estimating Influenza Vaccine Efficacy From Challenge and Community-based Study Data, Am J Epidemiol 168(12), 1343-52

Brauer F, van den Driessche P, Wu J (eds) 2008 Mathematical Epidemiology, Lecture Notes in Mathematics vol. 1945, Springer

Diekmann O, Heesterbeek JAP, Roberts MG, 2010 The construction of next-generation matrices for compartmental epidemic models, J R Soc Interface, 7:47, 873-885

Diekmann O, Heesterbeek JAP, Mathematical Epidemiology of Infectious Diseases: Model Building, Analysis and Interpretation, Wiley 2000

Eurostat 2006, Population Statistics 2006 edition, Official Publications of the European Communities (Luxembourg)

Knipl D, Röst G 2011 Modelling the strategies for age specific vaccination scheduling during influenza pandemic outbreaks, Math Biosci Eng 8(1), 123-139

Longini IM Jr, Halloran ME 2005 Strategy for distribution of influenza vaccine to high-risk groups and children, Am J Epidemiol bf 161(4), 303

Medlock J, Galvani AP 2009 Optimizing influenza vaccine distribution, Science 325(5948), 1705 - 1708

Moghadas SM, Bowman CS, Röst G, Wu J 2008 Population-wide emergence of antiviral resistance during pandemic influenza PLOS ONE 3(3) e1839

Moghadas SM, Bowman CS, Röst G, Fisman D, Wu J 2009 Post-exposure prophylaxis during pandemic outbreaks, BMC Medicine, 3(73)

Mossong J, Hens N, Nit M et al. 2008 Social contacts and mixing patterns relevant to the spread of infectious diseases, PLOS Medicine, 5(3):e74

Nichol KL. Efficacy/clinical effectiveness of inactivated influenza virus vaccines in adults 1998, In: Nicholson KG, Webster RG, Hay AJ, editors. Textbook of influenza. Malden, MA: Blackwell Science Ltd., 1998:358 -72

Appendix

Here we shortly introduce how to use Wolfram Mathematica commands for simulations. To solve a differential equation analytically, one may use the command 'DSolve'. Since most of the systems used in epidemiology are not analytically solvable, we mostly treat them numerically by using 'NDSolve'. An example is detailed here, however, one can find many other options in the Help Menu. Comments can be left with (* text *) in any Mathematica code.

```
Clear[s, i, r]
α = 1 / 3; β = 1.5;
```

```
(*first give the parameters used in the differential equations*)
sys = NDSolve[
  (*the differential equation system is labeled,
  here by 'sys', for further references*)
  {s'[t] == -β * s[t] * i[t],
   (*the differential equation system consists of 3 equations,
   which are divided from each other by commas*)
   i'[t] == β * s[t] * i[t] - α * i[t],
   r'[t] == α * i[t],
   i[0] == 0.001, r[0] == 0,
   s[0] == 1 - 0.001      (*initial values are given here*)
  },
  {s, i, r},
  (*variables are given here we solve the system for.*)
  {t, 0, 150}
  (*independent variable must be given with its domain*)
 ]
```

```
{{s → InterpolatingFunction[{{0., 150.}}, <>],
  i → InterpolatingFunction[{{0., 150.}}, <>],
  r → InterpolatingFunction[{{0., 150.}}, <>]}}
```

To plot functions on the screen the command 'Plot' can be used, see below. The Mathematica documentation center (see Help menu) provides comprehensive information about the several additional options available.

```
Plot[i[t] /. sys, {t, 0, 40}]
```



A very useful and convenient command to observe the outcomes of a model for varying input parameters is 'Manipulate'. By simply moving the sliders we can control as many parameters as we wish.

```
Manipulate[
 DynamicModule[{α, sys, s, i, r, s0, r0},
  s0 = 1 - i0; r0 = 0; α = 1 / l; (*the parameters i0,
  β and l are not given here, they will be varied on the sliders*)
  sys = NDSolve[
    {s'[t] == -β * s[t] * i[t],
     i'[t] == β * s[t] * i[t] - α * i[t],
     r'[t] == α * i[t],
     s[0] == s0, i[0] == i0, r[0] == r0}, {s, i, r}, {t, 0, 150}];
  Plot[{s[t] /. sys, i[t] /. sys, r[t] /. sys}, {t, 0, 30},
   (*several curves can be plotted on the same figure*)
```

```
   PlotStyle → {Darker[Green], Red, Blue}, PlotRange → {0, 1}]
  (*specification of the figure,
  for example the colors of the curves and the domain of the y-axis.*)
 ],
 {{i0, 0.001, "I₀: initially infected"}, 0, 0.005, 0.001,
  ImageSize → Tiny, Appearance → "Labeled"}, (*first the variable,
 the initial value and the title which will be displayed next to the slider
  is specified here in this order, then the range and the step size follow*)
 {{β, 1.5, "β: transmission rate"}, 0.5, 5, 0.5,
  ImageSize → Tiny, Appearance → "Labeled"},
 {{l, 3, "1/α: duration of the infection (in days)"},
  1, 7, 1, ImageSize → Tiny, Appearance → "Labeled"},
 Initialization :> (i0 = 0.001; Needs["PlotLegends`"]), SaveDefinitions → True
]
```